

CURSOR

THE TINY MICRO COMPUTER NEWS SERVICE

VOLUMN I

\$2.00 per Issue

FEBRUARY 1980

\$9.50 Semi-Annually

ISSUE 2

WINTER CONSUMER ELECTRONICS SHOW

On Friday, 4 January 1980, after a leisurely "free" breakfast at one of the Las Vegas casino's (I subsequently lost the "free" tickets and paid the inflated full price. Ah, a great start!), Tim Hays and I dropped into Bally's Hilton suite.

Bally in turn, dropped a grenade on us!

Bally had signed a "Letter of Intent" regarding the sale of Bally Consumer Product Division to Fidelity Electronics Ltd., Microprocessor Div. (keep in mind, that a "Letter of Intent" is not the same as a sale).

We spent the next several hours looking over a handful of prototype cartridges, only one of which (Dogpatch) was fully functional. We were amazed to find that Bally was showing a cassette tape program by W & W Software (Bio-Rhythm) which looks spectacular with its employment of constantly changing screen colors, but does not produce an accurate bio-rhythm. All of which points out that one must be very careful when buying mail-order software!

Upon departing the Bally suite, we went immediately to the Fidelity booth (they produce "Chess Challenger", etc.), where we found, understandably, a lack of knowledge regarding future plans. We were told by Fidelity personnel that all product development plans and decisions are made by one man, and one man only; Mr. Sidney Samole, President. We were informed that Mr. Samole was unavailable at the time, and would be departing for Europe

soon after the show.

We have, in the past, learned to accept press release information as just that; press release information! We at CURSOR feel strongly that you, the Arcade user, deserve to be told what information we have been able to garner.

1. Bally had fully intended from the start, to market and produce the "Add-on" unit.

2. Bally's plans were brought to a screeching halt by the FCC. The CURSOR staff cannot find fault with Bally's decision to delay production until the FCC made a definitive policy statement.

3. We at CURSOR have read and heard a considerable amount of inaccurate information being passed on as "inside" information. If you, the consumer, accept this information as anything other than wishful

thinking, we fear you might have a long wait. Consider the following:

If a manufacturer has developed software, debugged it, found it commercially viable, and has decided to mark it, the ROM cartridge would not reach you, the consumer, for six to nine months after the go-ahead had been given. There are many obstacles to consider:

A. The microcomputer industry is currently suffering a "chip lag", meaning chip production is running way behind demand.

B. Burn-in lag is between 6-10 weeks.

MAIL ADDRESS: P.O. BOX 266, NORTH HOLLYWOOD, CA 91603

BUSINESS ADDRESS: 6115 CLYBOURN, SUITE 25, NORTH HOLLYWOOD, CA 91606 • (213) 763-7701



C. Distribution lag (seems to take approximately 6 weeks from factory to local dealer).

I mention the above three points only so you may understand the time-lag every manufacturer faces. This time-lag has especially inhibited BALLY'S desire to produce as many different ROM cartridges in sufficient quantity, as there is demand. Add to these problems the inflexible determination of Bally not to produce an inferior cartridge or product.

All we have to do to appreciate the quality of programs Bally has produced, is go to your nearest game cartridge retailer and take a look at 10 of the ATARI game cartridges and compare them to Bally. The American way in many things is quantity, not quality. Bally has given us a touch of the European sense of merchandise design--QUALITY--(in capital letters). We at CURSOR have purchased every cartridge Bally has produced, and have found only one clunker: STAR BATTLE! Can Atari or Odyssey II say the same? For this, we must say "THANKS BALLY" (even if we haven't got the Add-on unit yet)!

If Fidelity does acquire Bally Consumer Products Division, we are sure that a company with a product track record such as Fidelity, will continue the superior product development lead that Bally has established.

ALTERATIONS

Subscriber Chuck Zellers suggests you try adding the following lines to "Electric Bill" printed in the Jan 80 issue (we agree)

```
121 CX=43;CY=16
141 CX=30;CY=16
161 CX=-71;CY=16
351 CX=12;CY=16
376 CX=-71;CY=0
```

INFO

ONE THING THAT HAS NAGGED ME FROM THE BEGINNING, WAS THE INABILITY OF THE TINY BASIC TO CLEAR ALL THE LETTER VARIABLES AT THE BEGINNING OF A PROGRAM USING A LOOP TO SAVE MEMORY. I FINALLY FIGURED IT OUT, TRY THE FOLLOWING. -FRED C.

```
10 FOR RM=20078T0 20128STEP 2;%(RM)=0;NEXT
      RM
```

This routine uses 32 Bytes-clears A-Z

PEEK N' POKE

By now, approximately 1,000 new subscribers should be muttering "Hey, what was he doing in the one-line program?". That's right, I was poking a zero in the RAM memory locations for variables A through Z. I used the variable "RM" which is used to store the remainder of a division. Tiny Basic uses integers only, and the remainder, if any, is stored in "RM". In this case, we used RM as a variable. A PEEK statement could look like this:

```
10 PRINT %(-24576)
      or
10 A=%(-24576);PRINT A
```

When using PEEK or POKE, keep in mind that these functions access two bytes of data at a time in a single decimal number format. For instance, using the last example listed above, "A" will equal 10 (this location is the beginning of the Text area). But when we access (-24574) we get 15681. If you divide that decimal number by 256, assign the product to "B", and the remainder (RM) to "C", the product will be: B=61;C=65. Reverse the order, convert from ASCII, and the result is "A=". Try this:

```
10 Z=-24574;A=%(Z);B=A÷256;C=RM
20 CLEAR ;PRINT #1,"%(",Z,")=",A,"=",;
      TV=C;TV=B
```

The above routine will give you the first 2 characters after the line number on line 10. Machine language is however, handled somewhat differently. The following is a routine that calls up the on-board clock:

```
1005 CLEAR
1010 M=20180;B=M;C=1080
1020 L=-43;GOSUB C
1030 L=12371;GOSUB C
1040 L=-13288;GOSUB C
1050 L=-13871;GOSUB C
1070 CALL(B);STOP
1080 % (M)=L;M=M+2;RETURN
```

MACHINE LANGUAGE

An explanation of the above program in machine language follows:

D5	Save Basic Pointer
FF	Call Subroutine
53	#82, Load Regs
30	Horizontal Screen Position
18	Vertical Screen Position
CC	BC Color, FC Color, Blank Existing Display, Character Size, Display both Minutes & Seconds w/colon separator.

D1 Restore Basic Pointer
C9 Go back to Basic

Next, we must convert the HEX program to decimal, first reversing their order, then converting the resulting hex number to decimal as follows:

FFD5=-43 (Line 1020)
3053=12371 (Line 1030)
CC18=-13288 (Line 1040)
C9D1=-13871 (Line 1050)

NOTE: When selecting an on-board sub-routine, list subroutine # as SR#+1 i.e., 82=83. 83 in Decimal=53 in Hex

This Clock routine does not increment or decrement. Would one of the professionals or "hot-shot" hackers get this thing running and send us the program to print?

HEX TO DECIMAL CONVERTER BY FRED CORNETT

```
9 PRINT "HEX # ?",
10 FOR A=1TO 4
11 @ (A)=KP
12 IF @ (A)>47IF @ (A)<58TV=@ (A);@ (A)=@ (A)-
48;NEXT A
35 IF @ (A)>64IF @ (A)<71TV=@ (A);@ (A)=@ (A)-
55;NEXT A
37 IF @ (1)<16GOTO 60
40 GOTO 20
60 B=4096;T=0;FOR A=1TO 4
65 IF A=1IF @ (A)>7GOSUB 100;NEXT A
70 IF A=1IF @ (A)<8@ (1)=@ (1)xB;T=T+@ (1);NE
XT A
75 B=B÷16;@ (A)=@ (A)xB;T=T+@ (A)
80 NEXT A;GOTO 150
100 T=-32767;IF @ (A)=8RETURN
110 T=T+((@ (A)-8)x4096)-1;RETURN
150 PRINT ;PRINT #1,"DEC. EQUIV=",T
160 GOTO 9
```

Remember to reverse Hex pair order prior to input.

STRING ARRAY @ (A) MEMORY LOCATOR

10 Z=(-24576)+(1800-SZ)+4

"Z" will equal the memory location of @ (1). To find successive string arrays, increment Z by 2 for each successive array you wish to find (Z=Z+2).

I trust that the forgoing information will keep many of you busy until the next issue comes out. Happy Headaches!!

PRINTER

The original Bally design of the audio Cassette interface was to include a 1/8" printer jack. By following the instructions below, you will have created a TTL Level (Voltage) RS232 Standard ASCII Serial Interface; 1 start bit, 8 data bits, 1 stop bit, no parity bit, no handshaking provision, at 300 Baud (30 CPS). When purchasing a printer you must specify "automatic line feed" as Bally Basic does not send line feeds. Use *PRINT to enable the printer, and :RETURN to disengage the printer.

NOTE: A modification may be necessary to your printer. Many printers use a line-receiver type IC which can be programmed for different input signal swings by tying a pin directly or through a resistor to a (-) voltage, to signal ground (0v), or to a (+) voltage. Consult your printer schematic and the IC manufacturer's data book as necessary.


PARTS LIST:

- 1-Jack (i.e., Switchcraft #41 or TR2A)
- 1-Drill bit (to drill through Aluminum for jack)
- Resin solder, soldering iron
- Phillips screwdriver
- Twisted strand hook-up wire.(2 different colors)
- Masking Tape
- Deburring tool or small file.(for cleaning up drilled hole)

STEP 1: Remove top cover of audio cassette interface by removing 2 bottom screws.

STEP 2: Remove PC board assembly by removing 4 screws.


STEP 3: Remove large plug and cables from end of board (if your unit is so equipped), if not, be careful not to bend connections severely or they may break.



STEP 4: Referring to the above drawing, mount the jack of your choice. Make sure jack does not interfere with the mounting of the PC board or make contact with any component or foil.

We suggest that you cover the audio cassette interface face-plate with a small piece of masking tape and center tap the

hole location prior to using your drill (remember the face-plate is aluminum, the drill bit will have a tendency to walk away from you, thereby scratching the face plate). Deburr hole as necessary.



STEP 5: Using 2 color stranded wire, twist two pieces together to form a twisted pair. Referring to the above PC board diagram, connect one end of one wire to the ground side of the jack and the other end (same wire) to pin 8 of IC5(MC14503B). Connect one end of the unused different colored wire to the high side of the jack and the other end to pin 7 of IC5.

STEP 6: Reassemble the unit reversing steps 1 through 3.

BUBBLE SORT

The following program will take a list of numbers, sort them, and reprint them in numeric order. When prompted "# OF ARRAYS?", input how many numbers you wish to sort, and then input the numbers, pressing "GO" at the end of each number. Our thanks go to Tim Hays for sharing his excellent program with us. HOW ABOUT YOU? Do you have a program that you think is sharp or interesting? PLEASE send it to CURSOR, and we'll share it with our readers.

```

1 .BUBBLE SORT
2 .USES 199 BYTES
3 .TIM HAYS 1/13/80
30000 CLEAR ;NT=1;INPUT "# OF ARRAYS ?"N;
CLEAR ;PRINT " INPUT DATA";FOR P=1TO
N;PRINT P;INPUT "="@(P);NEXT P
30010 CLEAR ;PRINT " SORTING..";I=N-1;FOR
J=1TO I;K=J+1;FOR L=NTO KSTEP -1;MU=
L;IF @ (L)>@ (J)GOTO 30030
30020 M=@ (L);@ (L)=@ (J);@ (J)=M
30030 NEXT L;NEXT J;PRINT " RESULT↓";FOR P
=1TO N;PRINT P,"=",@(P);NEXT P

```

CAMEL

This program has been plagiarized from "More Basic Computer Games", edited by David H. Ahl (\$7.50), pg 24. We have printed this program for the exclusive purpose of showing you how program conversions from one Basic to another are accomplished. We strongly suggest the purchase of this book to enable you to compare the statements. We have kept the same line numbers wherever possible. Word text has been changed extensively. When inputting this program, use as many control words as possible. SZ=21

SCENARIO:

The object is to travel 200 miles across the desert. You're being chased by a tribe of "stoned" pygmies. You have one canteen, which will last 6 drinks, it can be re-filled by finding an oasis. During your journey, you encounter hazards, i.e., crazed slavers, sand storms, vampire ants, etc. When prompted "INPUT COMMAND:", you may select from the following:

- 1=Drink from canteen.
- 2=Moderate speed ahead.
- 3=Ahead full speed
- 4=Stop for the night.
- 5>Status Check.
- 6=Hope for help.

```

10 CLEAR ;NT=6;FOR RM=20078TO 20128STEP
2;% (RM)=0;NEXT RM;S=6;Z=4
340 IF C>199GOTO 1210
350 Z=Z-1
355 IF Z=1PRINT "WARNING-DRINK"
360 IF Z<0GOTO 1630
370 P=P+1;D=(RND (100)+25)÷10
390 IF Q>0GOTO 940
400 IF P<4GOTO 470
410 B=B+D;IF B<CGOTO 460
430 PRINT "PYGMIES CAUGHT UP";PRINT "CAME
L & PEOPLE SOUP";GOTO 1560
460 PRINT #1,"PYGMIES-",C-B," MILES TO RE
AR
470 PRINT #1,"YOU'VE GONE ",C," MILES!
480 INPUT "INPUT COMMAND:"Y;CLEAR ;IF Y=1
GOTO 830
505 IF Y=2GOTO 610
510 IF Y=3GOTO 680
515 IF Y=4GOTO 760
520 IF Y=5GOTO 790
550 T=RND (10);IF T#1GOTO 1200
570 PRINT "HELP IS HERE!";PRINT "YOU ARE
ILL";S=3;Z=4;GOTO 340
610 F=F+1;IF F=8GOTO 1190
630 GOSUB 880;E=RND (10);C=C+E
660 PRINT "YOUR CAMEL LIKES THIS PACE";GO

```

```

TO 340
680 F=F+3; IF F>7 GOTO 1190
700 GOSUB 880; E=2xRND (10); C=C+E; PRINT "C
AMEL IS RACING OVER SANDS"; PRINT ; GOT
O 340
760 PRINT "REST AT LAST"; F=0; GOTO 350
790 PRINT #1, "CAMEL HAS ", 7-F, " DAYS TO G
O
800 PRINT #1, "YOU HAVE ", S, " DRINKS LEFT.
810 PRINT #1, "YOU CAN GO ", Z, " COMMANDS
811 PRINT "BEFOR DRINK
830 S=S-1; IF S<0 GOTO 1200
850 PRINT "LOOK FOR NEXT OASIS"; Z=4; GOTO
480
880 A=RND (100); IF A>5 GOTO 1120
900 PRINT "SLAVER GOT YOU"; PRINT "SHEIK T
O PAY RANSOM"; PRINT "LOOK FOR PYGMIES
!
940 INPUT "7=RUN 8=WAIT" X; IF X=8 GOTO 1060
990 E=RND (10); IF E<5 GOTO 1040
1010 PRINT "YAHOO!"; PRINT "YOU MADE IT"; Q=
0; GOTO 340
1040 PRINT "YOU WERE KILLED BY"; PRINT "VAM
PIRE ANTS"; GOTO 1410
1060 E=RND (100); IF E>24 GOTO 1100
1090 PRINT "RANSOM PAID"; PRINT "YOU ARE FR
EE!"; Q=0; GOTO 340
1100 PRINT "SHEIK TO COLLECT"; PRINT "WAIT!
"; GOTO 340
1120 A=RND (10); IF A>2 GOTO 1240
1140 PRINT "→OASIS← YOUR CAMEL IS
1141 PRINT "FILLING CANTEEN & EATING FIGS"
; Z=4; S=6; RETURN
1190 PRINT "SWINE-YOU RAN CAMEL TO "; PRINT
"DEATH
1200 GOTO 1410
1210 PRINT "YOU WON-A PARTY IS GIVEN TO H
ONOR YOU-PYGMIES WILL ATTEND!"; GOTO 1
560
1240 E=RND (100); IF E>5 GOTO 1350
1260 PRINT "SANDSTORM-PRAY!"; G=RND (10); H=
RND (10); IF H<5 GOTO 1315
1300 C=C+G; GOTO 1320
1315 C=C-G
1320 PRINT #1, "YOU ARE BLOWN ", G; PRINT "MI
LES OFF"; RETURN
1350 E=RND (100); IF E>5 RETURN
1390 PRINT "CAMEL HURT HUMP! LUCKY-"; PRINT
"PYGMIES WERE WEARY"; RETURN
1410 U=RND (10); PRINT "YOU DIED IN DESERT"
; IF U>2 GOTO 1460
1440 PRINT "CAMEL UNION NOT TO ATTEND FUNE
RAL!"; GOTO 1560
1460 IF U>4 GOTO 1490
1470 PRINT "YOUR BODY EATEN BY"; PRINT "AAR
DVARKS"; GOTO 1560
1490 IF U>6 GOTO 1520
1500 PRINT "SHEIK USING YOUR HEAD FOR PURS
E!"; GOTO 1560

```

```

1520 IF U>8 GOTO 1550
1530 PRINT "NEVER GOTO DESERT"; GOTO 1560
1550 PRINT "TURKEYS FLY"; PRINT "NOT RIDE C
AMELS
1560 A=KP; GOTO 10
1630 PRINT "OUT OF WATER"; GOTO 1410
~~~~~
```

STATEMENT OF POLICY

CURSOR was originally formulated with one intent; to alleviate the acute degree of frustration we, as users, were experiencing in the total lack of published hardware-interfacing information, along with a void of debugged, sophisticated, viable software(public domain).

In our pursuit of that goal, we have employed the majority of your subscription dollars in advertising, and upgrading the quality of this newsletter (as this issue will attest). We promised in our original promotional material that we would strive to remain aware of our readers demands. Recently it has come to our attention that over 80% of our subscribers lack the time and/or expertise to fabricate the hardware we are formulating, this creating a great demand for finished products.

We are now expanding our facilities to include ready-to-run products (Software & Hardware), which will in no way decrease the publication of all viable technical information. As an example, we plan to offer a "light pen" for sale, complete with software, in the near future. Also, we will publish a schematic and parts list, along with software instructions, for those capable of building their own.

We at CURSOR are not involved in a get-rich-quick scheme. Our pricing policy will keep the prices low enough to put CURSOR products in the hands of all those who desire them, at the same time delivering a product that is properly engineered with state-of-the-art technology. One reason we generally see a poor quality in the software being offered elsewhere, is the professional programmers in our midst are contractually eliminated from moonlighting, and cannot attach their name to software in fear of losing their jobs. By creating a CURSOR line of software, we are openly soliciting the professionals (and amateurs) to submit to CURSOR any software they have created for their own enjoyment or use. Should it be selected for inclusion in our "Software of the Month" tape, they will share in the profits (no mention of their

names). This will create a considerable boon to the software purchaser. CURSOR guarantees "CURSOR" software as being original, innovative, glitch-free, and inexpensive.

We welcome your criticism of CURSOR policy or format!

USER GROUP DATA

CURSOR is committed to the formation of User Groups throughout the USA. If you have a user group meeting, or wish to start one, send us the particulars, and we'll help get you started.

MICHIGAN:

BUGS(Bally Users Group)
Saturday, 12 April 1980
12:00 Noon

George Moses Co.
110 East North Street
Brighton, Michigan
Phone (313)227-1575

LOS ANGELES AREA:

Wednesday, 6 March 1980
7:30 PM

Fred Cornett
(213) 763-7701
Call for directions

MEMORY MAP

DECIMAL

On Board ROM	Ø - 8191
Bally Basic ROM	8192 - 12287
Screen Memory Area	16384 - Ø479
Bally Basic Graphics/ Program area	16384 - 19983
Bally Basic Scratchpad	ØØØØ - Ø463
Tape Input Buffer	ØØØ2 - ØØ49
Variables begin at	ØØ78
Line Input Buffer (Ø4 Characters)	ØØ18Ø - ØØ283
Stack Area	ØØ284 - ØØ462
Text Area	-24576 --22777
Note Lookup Table	12Ø46

BACK ISSUES ETC.

Those desiring the January CURSOR may send \$1.60 to our mailing address, and receive it shortly after we receive it(Jan covers adding full size ASCII Keyboard).

ISSUES ON TAPE

We have received many requests for a tape containing all the programs in an issue. For those individuals desirous of saving the time involved in the input of our programs, we offer tapes of each issue at \$6.75 (time consuming for us). Please specify Issue #, and allow approx. 3 weeks for delivery.


WUMPUS

BY CHRIS PRUITT & ROBERT LEAKE

RULES: The Wumpus, a mythical creature which no man has seen before, lives in a cavern of 20 rooms. Each room connects to 3 other rooms.

Your mission is to deduce which room the Wumpus is sleeping in (it usually sleeps), and shoot it with an arrow. You have 5 arrows. You fire an arrow by inputting "Ø" when prompted "MOVE TO?". You can then shoot in up to 5 rooms. The rooms must be joined to each other, or you can ricochet and be killed.

There are other hazards. Two rooms have Bats and 2 have Pits. Also, the Wumpus moves when you wake him, and he may eat you.



CAVERN DIAGRAM

A Squashed
Dodecahedron

LINE EXPLANATION:

- 1Ø- 7Ø Sets up Bats, Pits, Wumpus & Player
- 8Ø- 9Ø Checks for Pits & falling in
- 1ØØ-11Ø Checks for Bats & carrying off
- 14Ø-15Ø Checks for nearby Pits
- 16Ø-17Ø Checks for nearby Bats
- 18Ø Checks for nearby Wumpus
- 19Ø Prints Room # & looks for Wumpus
- 2ØØ-24Ø 1 in 4 chance Wumpus eats you
- 25Ø-28Ø Wumpus' move
- 29Ø Connecting rooms
- 30Ø-34Ø Your room choice
- 35Ø-45Ø Arrow-rooms & distance
- 46Ø-5ØØ 1 in 4 chance Arrow gets you
- 51Ø-53Ø Arrow-checks to see if you got Wumpus
- 54Ø-56Ø If Arrow misses Wumpus, he moves
- 57Ø Remaining Arrows
- 59Ø-61Ø Start game over with/without same setup
- 62Ø-65Ø Subroutine for "Bat-Snatch"
- 67Ø-79Ø Room adjacency

NOTE: The rules we have printed are purposely somewhat sketchy. We leave the discovery up to you.

```

1 .WUMPUS
10 U=RND (20)
20 V=RND (20); IF V=UGOTO 20
30 P=RND (20); IF (P=U)+(P=V) GOTO 30
40 Q=RND (20); IF (Q=U)+(Q=V)+(Q=P) GOTO 40
50 S=RND (20); IF (S=U)+(S=V)+(S=P)+(S=Q) GOTO 50
60 T=RND (20); IF (T=U)+(T=V)+(T=P)+(T=Q)+(T=S) GOTO 60
70 Y=U; W=V; M=5
80 IF (Y=P)+(Y=Q)=0 GOTO 100
90 CLEAR ; FC=7; BC=0; CX=-30; CY=0; PRINT "FE
    LL IN PIT"; GOTO 590
100 IF Y#SIF Y#TGOTO 120
110 IF (Y=S)+(Y=T) GOSUB 620; GOTO 80
120 GOSUB 670
130 CLEAR
140 N=(P=A)+(P=B)+(P=C)+(Q=A)+(Q=B)+(Q=C);
    IF N=0 GOTO 160
150 FOR I=1 TO N; PRINT " I FEEL A DRAFT!";
    NEXT I
160 N=(S=A)+(S=B)+(S=C)+(T=A)+(T=B)+(T=C);
    IF N=0 GOTO 180
170 FOR I=1 TO N; PRINT " I HEAR WINGFLAPS!
    "; NEXT I
180 IF (W=A)+(W=B)+(W=C) PRINT " I SMELL A
    WUMPUS!"
190 PRINT ; PRINT " YOU'RE IN ROOM", #3, Y; I
    F Y#WGOTO 290
200 PRINT ; PRINT " AND SO IS THE WUMPUS!
210 L=RND (4); FOR I=1 TO 700; NEXT I; IF L<4 GOTO 250
220 FOR I=1 TO 15; CLEAR; FC=99; BC=103
230 CX=RND (118)-69; CY=8xRND (9)-40; PRINT
    "CHOMP"; NEXT I
240 CLEAR ; CY=0; PRINT " WUMPUS ATE YOU!";
    GOTO 590
250 IF L=1W=A
260 IF L=2W=B
270 IF L=3W=C
280 PRINT " BUT HE'S MOVING OFF!
290 PRINT ; PRINT " TUNNELS GO TO ", #3, A.#
    3, B, #3, C
300 PRINT ; INPUT " MOVE TO?" D
310 IF D=A Y=A; GOTO 80
320 IF D=B Y=B; GOTO 80
330 IF D=C Y=C; GOTO 80
340 IF DGOTO 300
350 K=Y; INPUT " HOW FAR?" E
360 IF E>5 GOTO 350
370 FOR I=1 TO E
380 INPUT " ROOM?" J
390 @I=J
400 NEXT I
410 FOR I=1 TO E
420 GOSUB 670
430 IF @I=A Y=A; GOTO 500
440 IF @I=B Y=B; GOTO 500
450 IF @I=C Y=C; GOTO 500
460 O=RND (4); IF O=1 Y=A
470 IF O=2 Y=B
480 IF O=3 Y=C
490 PRINT " BOINNG!
500 IF Y=KCLEAR ; FC=99; BC=0; CX=-36; CY=0; PR
    INT "ARROW GOT YOU!"; GOTO 590
510 IF Y=WCLEAR ; FC=99; CX=-30; CY=0; PRINT "
    YOU GOT HIM!"; GOTO 590
520 NEXT I
530 Y=W; GOSUB 670
540 R=RND (3); PRINT " YOU MISSED!"; IF R=1
    W=A
550 IF R=2 W=B
560 IF R=3 W=C
570 Y=K; M=M-1; IF M=0 CLEAR ; FC=176; BC=135; C
    X=-17; CY=0; PRINT " NO AMMO!"; GOTO 590
580 GOTO 120
590 FOR J=1 TO 999; NEXT J; CLEAR ; FC=0; BC=7;
    CY=0; INPUT " SAME SET UP? (Y=1, N=0) " H
600 IF H=1 GOTO 70
610 GOTO 10
620 CLEAR ; FC=7; BC=8; PRINT " ZAP!
630 FOR I=1 TO 15; CLEAR ; CX=RND (123)-69; CY
    =RND (9)-40; PRINT "FLAP"; NEXT I; CLEAR
640 CY=0; PRINT " SUPER BAT SNATCH!"; FOR
    I=1 TO 900; NEXT I; CLEAR
650 Y=RND (20); FC=0; BC=7
660 RETURN
670 IF Y>5 GOTO 710
680 A=Y+1; IF Y=5 A=1
690 B=Y-1; IF Y=1 B=5
700 C=Y+5; RETURN
710 IF Y>10 GOTO 740
720 A=Y+5; B=Y+4; IF Y=6 B=15
730 C=Y-5; RETURN
740 IF Y>15 GOTO 770
750 A=Y-4; IF Y=15 A=6
760 B=Y-5; C=Y+5; RETURN
770 A=Y+1; IF Y=20 A=16
780 B=Y-1; IF Y=16 B=20
790 C=Y-5; RETURN

```

VARIABLE INDEX

- A, B, C Place holders
- D Move to?
- E How far?
- H Same setup?
- I Counters
- J Arrows go through
- K Sub for Y
- L RND for Wumpus movement
- M Arrows
- N Counter for warning signs
- O Arrow ricochet
- P, Q Pits
- R RND Wumpus move w/arrow
- S, T Bats
- U Pre-You
- V Pre-Wumpus
- W Wumpus
- Y You
- F, G, X, Z Unused

CLASSIFIEDS

CURSOR SOFTWARE OF-THE MONTH


The following two photographs and descriptions make up the first CURSOR Cassette Tape Software offering. Both programs are listed on one tape, and include complete documentation. Price is \$8.95 and includes postage and all fees.

PROGRAM 1-A: MOON LANDING

Code: Basic/Machine Language

You're in the L.E.M. waiting for the instruction to break away from the mother ship. Once you do, you have to quickly scout for a safe landing spot. You carefully maneuver into a safe landing position; watching the drift (horiz.) and speed (vert.). If you successfully land, you have to gather a group of moon rocks before the countdown to blast-off for re-connection to the mother ship prior to running out of time and fuel. (Software selectable gravity wells.)

Program is partly in machine language to generate the fast acting user defined characters (L.E.M. does not blink) for: Horizontal LEM, LEM banked to the right, LEM banked to the left, 2 explosions (moving). Superb graphics & sound.




Look for CURSOR & SEBREE'S at the West Coast Computer Faire - San Francisco Civic Auditorium 14-16 March Booth #41

PROGRAM 1-B: BIO-RHYTHM

Through this computerized study of biological clocks you can predict physical, emotional and intellectual behavior at peak and critical periods. Bio-rhythm has helped U.S. airlines avoid crashes and athletes to choose their best competitive days; it has reduced dramatically the auto accident rate in Japan and increased the performances of sales forces, teachers and factory workers.

Bio-rhythm can help you predict outbreaks of illness, mental depression, days of tireless energy, best times for creative work, peak periods of mental and emotional control.

Very accurate graph format allows you to select and see your critical days individually.



Those of you desirous of a safe source of software on cassette tape CURSOR recommends SEBREE'S COMPUTING, 456 Granite, Monrovia, CA. 91016 (a letter will get you a catalog)

DIGITRENDS, Inc., 1813 East 12th St., Cleveland, Ohio 44114 (216)241-1813
Carries the full Bally line and says they "will ship anywhere"!